The D Programming Language Specification

D is a small imperative language. It has 32-bit integer constants, variables, and functions, and a minimal set of statements: assignment, if, while, and return. There are no modules, classes, objects, or external variables. There are no header files or libraries, but there are two predefined functions, get and put, which provide the ability to read and write integer values. Syntactically it is much like C. A program consists of a sequence of function definitions.

Grammar

Here is the basic grammar for D.

1. \( \text{program} ::= \text{functionDefinition} | \text{program functionDefinition} \)
2. \( \text{functionDefinition} ::= \text{int id ( ) }\{ \text{statements} \} \)
   \| \( \text{int id ( parameters ) }\{ \text{statements} \} \)
   \| \( \text{int id ( ) }\{ \text{declarations statements} \} \)
   \| \( \text{int id ( parameters ) }\{ \text{declarations statements} \} \)
3. \( \text{parameters ::= parameter | parameters , parameter} \)
4. \( \text{parameter ::= int id} \)
5. \( \text{declarations ::= declaration | declarations declaration} \)
6. \( \text{declaration ::= int id ;} \)
7. \( \text{statements ::= statement | statements statement} \)
8. \( \text{statement ::= }\{ \text{statements} \} | \text{id }= \text{expr ;} \)
   \| \( \text{if ( boolExpr ) statement | if ( boolExpr ) statement else statement} \)
   \| \( \text{while ( boolExpr ) statement | return expr ;} \)
9. \( \text{expr ::= term | expr + term | expr - term} \)
10. \( \text{term ::= factor | term * factor} \)
11. \( \text{factor ::= int | ( expr ) | id | id ( ) } | \text{id ( exprList )} \)
12. \( \text{exprList ::= expr | exprList , expr} \)
13. \( \text{boolExpr ::= relExpr | ! ( relExpr )} \)
14. \( \text{relExpr ::= expr == expr | expr > expr} \)

See next page for additional productions that are rewrites of the above list.
Rewritten productions to fix left recursion problems.

Instead of 1, use 15 and 16.

15. \( \text{program} ::= \text{functionDefinition} \text{ programTail} \text{ EOF} \)
16. \( \text{programTail} ::= \text{functionDefinition} \text{ programTail} | \epsilon \)

Instead of 3, use 17 and 18.

17. \( \text{parameters} ::= \text{parameter} \text{ parametersTail} \)
18. \( \text{parametersTail} ::= , \text{parameter} \text{ parametersTail} | \epsilon \)

Instead of 5, use 19 and 20.

19. \( \text{declarations} ::= \text{declaration} \text{ declarationsTail} \)
20. \( \text{declarationsTail} ::= \text{declaration} \text{ declarationsTail} | \epsilon \)

Instead of 7, use 21 and 22.

21. \( \text{statements} ::= \text{statement} \text{ statementsTail} \)
22. \( \text{statementsTail} ::= \text{statement} \text{ statementsTail} | \epsilon \)

Instead of 9, use 23 and 24.

23. \( \text{expr} ::= \text{term} \text{ exprTail} \)
24. \( \text{exprTail} ::= + \text{term} \text{ exprTail} | - \text{term} \text{ exprTail} | \epsilon \)

Instead of 10, use 25 and 26.

25. \( \text{term} ::= \text{factor} \text{ termTail} \)
26. \( \text{termTail} ::= * \text{factor} \text{ termTail} | \epsilon \)

Instead of 12, use 27 and 28.

27. \( \text{exprList} ::= \text{expr} \text{ exprListTail} \)
28. \( \text{exprListTail} ::= , \text{expr} \text{ exprListTail} | \epsilon \)

It is also possible to replace the tail recursive statements above with iterations. For example:

\[
\begin{align*}
\text{parameters} &::= \text{parameter} (, \text{parameter})* \\
\text{declarations} &::= \text{declaration} (\text{declaration})* \\
\end{align*}
\]
Features implemented in the Scanner class

There are two undefined nonterminals in the grammar: id and int. An integer, int, consists of 1 or more digits (0-9) and denotes a decimal integer. An identifier, id, must begin with a letter, and consists of 1 or more letters, digits, and underscores. Upper- and lower-case letters are distinct, thus aa, AA, Aa, and aA are four different identifiers.

Comments, blanks, and other whitespace are ignored except as needed to separate adjacent syntactic tokens. A comment begins with the token // and continues to the end of the line.

The keywords in the grammar (int, if, etc.) are reserved and may not be used as identifiers.

General comments

A program consists of one or more function definitions.

All functions should have distinct names, and one of the functions must be named main (not enforced). A program is executed by evaluating main().

All integer values are 32-bit, two's complement numbers (not enforced).

D includes binary arithmetic operators +, -, and *. There is no division operator or unary + or - operators. The value -n can be computed by evaluating 0-n.

A boolExpr is a logical expression, which may only be used as a condition in an if or while statement. Logical expressions do not have integer values and cannot be stored in variables.

In conditional statements, each else is paired with the nearest previous unpaired if.

All local variables must be declared at the beginning of a function, and each declaration introduces a single variable. The local variables and parameters in a function must have distinct names (not enforced), and their scope extends over the entire function definition.

All functions are integer-valued, including main.

Function execution must terminate by executing a return statement. It is an error to "fall off" the end of the list of statements that make up a function body (not enforced).

There are two predefined functions that provide integer input and output (not implemented).

get() yields the next integer value from the standard input. If the next non-whitespace characters in the input do not form an integer constant, execution of get() is not defined.

put(x) yields the value of x and, as a side effect, prints that value on the next line of the standard output. Like all function calls, put(x) is an expression, so it may not be used as a statement. The statement x=put(x); may be used to print the value of a variable.